// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/math/SafeMath.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (utils/math/SafeMath.sol)

pragma solidity ^0.8.0;

// CAUTION
// This version of SafeMath should only be used with Solidity 0.8 or later,
// because it relies on the compiler's built in overflow checks.

/**
 * @dev Wrappers over Solidity's arithmetic operations.
 *
 * NOTE: `SafeMath` is generally not needed starting with Solidity 0.8, since the compiler
 * now has built in overflow checking.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            uint256 c = a + b;
            if (c < a) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b > a) return (false, 0);
            return (true, a - b);
        }
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
            // benefit is lost if 'b' is also tested.
            // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
            if (a == 0) return (true, 0);
            uint256 c = a * b;
            if (c / a != b) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the division of two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a / b);
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a % b);
        }
    }

    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        return a + b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return a - b;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        return a * b;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator.
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return a % b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {trySub}.
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b <= a, errorMessage);
            return a - b;
        }
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a / b;
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting with custom message when dividing by zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryMod}.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a % b;
        }
    }
}

// File: contracts/contracts-BatchDepositWithKeyRegistry/BatchDepositWithKeyRegistry.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity  ^0.8.13;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/utils/math/SafeMath.sol";
import "solidity-bytes-utils/contracts/BytesLib.sol";

import "./interfaces/IBatchDepositWithKeyRegistry.sol";
import "./interfaces/IDepositContract.sol";
import "./interfaces/IKeyRegistry.sol";

contract BatchDepositWithKeyRegistry is IBatchDepositWithKeyRegistry, Ownable, ReentrancyGuard{
    using SafeMath for uint256;

    /// @dev asset manager address
    address public assetManager;

    /// @dev ETH2 depositContract address
    address public immutable depositContract;

    /// @dev KeyRegistry contract address
    address public keyRegistry;

    uint256 constant public DEPOSIT_AMOUNT = 32 ether;
    uint256 internal constant DEPOSIT_AMOUNT_UNIT = 1000000000 wei;
    uint256 constant public SIGNATURE_LENGTH = 96;

    /// @dev default value for maximum number of Ethereum 2.0 validators registered in a single  call 
    uint256 public maxDepositsPerCall;

    /// @dev deposit whitelist
    mapping(address => bool) public whitelist;

    /// @dev depositETH method switch
    bool public depositEnabled;

    constructor(
        address _assetManager,
        address _depositContract,
        address _keyRegistry,
        uint256 _initMaxDepositCount
    ) {
        require(address(_assetManager) != address(0), "invalid init asset manager address");
        require(address(_depositContract) != address(0), "invalid init ETH2 depositContract address");
        require(address(_keyRegistry) != address(0), "invalid init KeyRegistry address");
        require(_initMaxDepositCount != 0, "invalid init max deposit per call");
        assetManager = _assetManager;
        depositContract = _depositContract;
        keyRegistry = _keyRegistry;
        maxDepositsPerCall = _initMaxDepositCount;

        emit Deployed(msg.sender,_assetManager, _depositContract, _keyRegistry, _initMaxDepositCount);
    }

    /// @dev receive ETH and deposit
    receive() external payable {
        depositETH();
    }

    /// @dev asset manager authority
    modifier onlyAssetManager() {
        require(msg.sender == assetManager, "not asset manager");
        _;
    }

    /// @dev whitelist users` authority
    modifier onlyWhitelisted() {
        require(whitelist[msg.sender] == true, "not in the whitelist");
        _;
    }

    /// @dev depositETH method switch
    modifier onlyDepositEnabled() {
        require(depositEnabled == true, "not enabled");
        _;
    }

    /// @dev Turn on the depositETH switch
    function enable() public onlyOwner {
        require(depositEnabled != true, "already enabled");
        depositEnabled = true;
        emit Enable();
    }

    /// @dev Turn off the depositETH switch
    function pause() public onlyOwner {
        require(depositEnabled != false, "already paused");
        depositEnabled = false;
        emit Pause();
    }

    /// @dev update asset manager address, only owner can operator
    /// @param _newAssetManager new asset manager address
    function updateAssetManager(address _newAssetManager) public onlyOwner {
        require(address(_newAssetManager) != address(0), "invalid new asset manager address");
        require(address(_newAssetManager) != address(assetManager), "asset manager address repeat");
        assetManager = _newAssetManager;
        emit UpdateAssetManager(_newAssetManager);
    }

    /// @dev update maxDepositsPerCall, only asset manager can operator
    /// @param _newMaxDepositCount new maxDepositsPerCall
    function updateMaxDepositCount(uint256 _newMaxDepositCount) public onlyAssetManager {
        require(_newMaxDepositCount != 0, "zero max deposit count set");
        require(_newMaxDepositCount != maxDepositsPerCall, "repeat setting max deposit count");                                                                                  
        maxDepositsPerCall = _newMaxDepositCount;
        emit UpdateMaxDeposit(_newMaxDepositCount);
    }

    /// @dev update KeyRegistry contract address, only owner can operator
    /// @param _newKeyRegistry new KeyRegistry contract address
    function updateKeyRegistry(address _newKeyRegistry) public onlyOwner {
        require(address(_newKeyRegistry) != address(0), "invalid new KeyRegistry address");
        require(address(_newKeyRegistry) != address(keyRegistry), "KeyRegistry address repeat");
        keyRegistry = _newKeyRegistry;
        emit UpdateKeyRegistry(_newKeyRegistry);
    }

    // @dev add single account or multiple accounts to whitelist
    /// @param _addrs accounts addresses
    function addAddressesToWhitelist(address[] memory _addrs) public onlyAssetManager {
        for (uint256 i = 0; i < _addrs.length; i++) {
            require(_addrs[i] != address(0),"invalid address");
            require(whitelist[_addrs[i]] != true, "already in the whitelist");
            whitelist[_addrs[i]] = true;
            emit WhitelistAddressAdded(_addrs[i]);
        }
    }

    /// @dev remove single accounts or multiple accounts from whitelist
    /// @param _addrs accounts addresses
    function removeAddressesFromWhitelist(address[] memory _addrs) public onlyAssetManager {
        for (uint256 i = 0; i < _addrs.length; i++) {
            require(_addrs[i] != address(0),"invalid address");
            require(whitelist[_addrs[i]] == true, "not in the whitelist");
            delete whitelist[_addrs[i]];
            emit WhitelistAddressRemoved(_addrs[i]);
        }
    }

    /// @dev deposit ETH to ETH2 depositContract, only in the whitelist can use this method
    function depositETH() public payable nonReentrant onlyWhitelisted onlyDepositEnabled {
        uint256 ethAmount = msg.value;
        uint256 numKeys = ethAmount.div(DEPOSIT_AMOUNT);
        require(ethAmount != 0, "zero ETH deposited");
        require(ethAmount.mod(DEPOSIT_AMOUNT) == 0, "not a multiple of 32");
        require(numKeys <= maxDepositsPerCall, "over max keys");

        eth2Deposit(numKeys);

        emit ETHDeposited(msg.sender, numKeys, ethAmount);
    }

    function eth2Deposit(uint256 numDeposit) internal {
        (
            bytes[] memory pubKeys,
            bytes[] memory withdrawalCredentials,
            bytes[] memory signatures
        ) = IKeyRegistry(keyRegistry).assignKeys(msg.sender, numDeposit);

        uint256 keyAmount = pubKeys.length;
        require(keyAmount != 0, "error pubKeys amounts");
        require(withdrawalCredentials.length == keyAmount, "error withdrawalCredentials amounts");
        require(signatures.length == keyAmount, "error signatures amounts");

        for (uint256 i = 0; i < keyAmount; i++) {
            _deposit(pubKeys[i], withdrawalCredentials[i], signatures[i]);
        }
    }

    function _deposit(bytes memory pubKey, bytes memory withdrawalCredential, bytes memory signature) internal {
        uint256 valueAmount = DEPOSIT_AMOUNT;
        uint256 depositAmounts = valueAmount.div(DEPOSIT_AMOUNT_UNIT);
        require(depositAmounts.mul(DEPOSIT_AMOUNT_UNIT) == valueAmount, "depositAmounts error");

        bytes32 pubkeyRoot = sha256(_pad64(pubKey));
        bytes32 signatureRoot = sha256(
            abi.encodePacked(
                sha256(BytesLib.slice(signature, 0, 64)),
                sha256(_pad64(BytesLib.slice(signature, 64, SIGNATURE_LENGTH.sub(64))))
                )
            );
        
        bytes32 depositDataRoot = sha256(
          abi.encodePacked(
              sha256(abi.encodePacked(pubkeyRoot,withdrawalCredential)),
              sha256(abi.encodePacked(_toLittleEndian64(depositAmounts),signatureRoot))
          )
        );
        uint256 targetBalance = address(this).balance.sub(valueAmount);

        IDepositContract(depositContract).deposit{value : valueAmount}(pubKey, abi.encodePacked(withdrawalCredential), signature, depositDataRoot);
        require(address(this).balance == targetBalance, "expecting deposit to happen");
        emit Deposited(msg.sender, pubKey, withdrawalCredential, signature, depositDataRoot, valueAmount);
    }

    /// @dev sweep ETH in this contract
    /// @param _receiver ETH receiver address
    function sweep(address _receiver) public onlyOwner nonReentrant {
        uint256 currentBalance = address(this).balance;
        require(currentBalance != 0, "invalid balance");
        require(_receiver != address(0), "invalid receiver address");
        (bool sent,) = payable(_receiver).call{value: currentBalance}("");
        require(sent, "failed sweep");

        emit Swept(msg.sender, _receiver, currentBalance);
    }

    function _pad64(bytes memory _b) internal pure returns (bytes memory) {
        assert(_b.length >= 32 && _b.length <= 64);
        if (64 == _b.length)
            return _b;

        bytes memory zero32 = new bytes(32);
        assembly { mstore(add(zero32, 0x20), 0) }

        if (32 == _b.length)
            return BytesLib.concat(_b, zero32);
        else
            return BytesLib.concat(_b, BytesLib.slice(zero32, 0, uint256(64).sub(_b.length)));
    }

    function _toLittleEndian64(uint256 _value) internal pure returns (uint256 result) {
        result = 0;
        uint256 temp_value = _value;
        for (uint256 i = 0; i < 8; ++i) {
            result = (result << 8) | (temp_value & 0xFF);
            temp_value >>= 8;
        }

        assert(0 == temp_value);    // fully converted
        result <<= (24 * 8);
    }
}
// File: contracts/contracts-BatchDepositWithKeyRegistry/interfaces/IBatchDepositWithKeyRegistry.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity  ^0.8.13;

interface IBatchDepositWithKeyRegistry {

    event Deployed(
        address indexed deployer,
        address indexed assetManager, 
        address indexed depositContract,
        address keyRegistry,
        uint256 initMaxDepositCount
    );

    event ETHDeposited(address from, uint256 keysNumber, uint256 ethAmount);

    event Enable();

    event Pause();

    event UpdateAssetManager(address newAssetManager);

    event UpdateMaxDeposit(uint256 newMaxDepositCount);

    event UpdateKeyRegistry(address newKeyRegistry);

    event WhitelistAddressAdded(address userAddress);

    event WhitelistAddressRemoved(address userAddress);

    event Deposited(
        address indexed from, 
        bytes pubkey, 
        bytes withdrawalCredentials, 
        bytes signature,
        bytes32 depositDataRoot,
        uint256 ethAmount
    );

    event Swept(address from, address to, uint256 amount);

    function whitelist(address) external view returns(bool);

    function enable() external;

    function pause() external;

    function updateAssetManager(address _newAssetManager) external;

    function updateKeyRegistry(address _newKeyRegistry) external;

    function addAddressesToWhitelist(address[] memory _addrs) external;

    function removeAddressesFromWhitelist(address[] memory _addrs) external;

    function depositETH() external payable;

    function sweep(address _receiver) external;
}
// File: contracts/contracts-BatchDepositWithKeyRegistry/interfaces/IDepositContract.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.0;

interface IDepositContract {
    /// @notice A processed deposit event.
    event DepositEvent(
        bytes pubkey,
        bytes withdrawal_credentials,
        bytes amount,
        bytes signature,
        bytes index
    );

    function deposit(
        bytes calldata pubkey,
        bytes calldata withdrawal_credentials,
        bytes calldata signature,
        bytes32 deposit_data_root
    ) external payable;

    function get_deposit_root() external view returns (bytes32);

    function get_deposit_count() external view returns (bytes memory);
}
// File: contracts/contracts-BatchDepositWithKeyRegistry/interfaces/IKeyRegistry.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.13;


interface IKeyRegistry {

    /// @dev Emitted when BatchDeposit contract address updated
    event UpdateBatchDepositContract(address indexed newBatchDepositContract);

    /// @param _newBatchDepositContract new BatchDeposit contract address
    function updateBatchDepositContract(address _newBatchDepositContract) external;

    /// @dev Emitted when Asset Manager updated
    event UpdateAssetManager(address indexed _assetManager);

    /// @param _assetManager assetManager address
    function updateAssetManager(address _assetManager) external;


    /// @dev Key structure
    struct KeyInfo {
        bytes pubkey;
        bytes withdrawalCredential;
        bytes signature;
    }

    /// @dev Emitted when one key added
    event AddKey(bytes pubkey);

    /// @dev AssetManager submit keys. pubkeys, withdrawalCredentials and signatures are one-to-one correspondence in array.
    /// @param _addr address that can use keys submitted
    /// @param quantity keys' quantity
    /// @param pubkeys pubkeys
    /// @param withdrawalCredentials withdrawalCredentials
    /// @param signatures signatures
    function addKeys(
        address _addr,
        uint256 quantity,
        bytes[] calldata pubkeys,
        bytes[] calldata withdrawalCredentials,
        bytes[] calldata signatures
    ) external;

    /// @dev Emitted when one key removed
    event RemoveKey(bytes pubkey);

    /// @dev AssetManager remove keys. Will remove [startIdx, endIdx] range in keys[_addr].
    /// @param _addr address's keys to remove
    /// @param startIdx start index to remove, the startIdx element is removed
    /// @param endIdx end index to remove, the endIdx element is removed.
    function removeKeys(
        address _addr,
        uint256 startIdx,
        uint256 endIdx
    ) external;

    /// @dev Emitted when one key assigned
    event AssignKey(bytes pubkey);

    /// @dev EPotter assign keys to use. require keys are enough.
    /// @param _addr address's keys to assign
    /// @param quantity quantity
    /// @return return pubkeys, withdrawalCredentials and signatures, they are one-to-one correspondence in array.
    function assignKeys(address _addr, uint256 quantity)
        external
        returns (
            bytes[] memory,
            bytes[] memory,
            bytes[] memory
        );

    /// @dev Get total quantity of _addr's keys
    /// @param _addr who's key
    /// @return total quantity
    function getTotalQuantity(address _addr) external view returns (uint256);

    /// @dev Get used quantity of _addr's keys
    /// @param _addr who's key
    /// @return used quantity
    function getUsedQuantity(address _addr) external view returns (uint256);

    /// @dev Get unused quantity of _addr's keys
    /// @param _addr who's key
    /// @return unused quantity
    function getUnusedQuantity(address _addr) external view returns (uint256);

    /// @dev get keys from startIdx to endIdx
    /// @param _addr who's key
    /// @param startIdx start index
    /// @param endIdx end index
    /// @return from start index to en index, keys' info
    function getMultipleKeys(address _addr, uint256 startIdx, uint256 endIdx) external view returns(KeyInfo[] memory);

    /// @dev get pointed key
    /// @param _addr who's key
    /// @param _index index
    function getSingleKey(address _addr, uint256 _index) external view returns(KeyInfo memory);
}

// File: solidity-bytes-utils/contracts/BytesLib.sol



// SPDX-License-Identifier: Unlicense
/*
 * @title Solidity Bytes Arrays Utils
 * @author Gonçalo Sá <goncalo.sa@consensys.net>
 *
 * @dev Bytes tightly packed arrays utility library for ethereum contracts written in Solidity.
 *      The library lets you concatenate, slice and type cast bytes arrays both in memory and storage.
 */
pragma solidity >=0.8.0 <0.9.0;


library BytesLib {
    function concat(
        bytes memory _preBytes,
        bytes memory _postBytes
    )
        internal
        pure
        returns (bytes memory)
    {
        bytes memory tempBytes;

        assembly {
            // Get a location of some free memory and store it in tempBytes as
            // Solidity does for memory variables.
            tempBytes := mload(0x40)

            // Store the length of the first bytes array at the beginning of
            // the memory for tempBytes.
            let length := mload(_preBytes)
            mstore(tempBytes, length)

            // Maintain a memory counter for the current write location in the
            // temp bytes array by adding the 32 bytes for the array length to
            // the starting location.
            let mc := add(tempBytes, 0x20)
            // Stop copying when the memory counter reaches the length of the
            // first bytes array.
            let end := add(mc, length)

            for {
                // Initialize a copy counter to the start of the _preBytes data,
                // 32 bytes into its memory.
                let cc := add(_preBytes, 0x20)
            } lt(mc, end) {
                // Increase both counters by 32 bytes each iteration.
                mc := add(mc, 0x20)
                cc := add(cc, 0x20)
            } {
                // Write the _preBytes data into the tempBytes memory 32 bytes
                // at a time.
                mstore(mc, mload(cc))
            }

            // Add the length of _postBytes to the current length of tempBytes
            // and store it as the new length in the first 32 bytes of the
            // tempBytes memory.
            length := mload(_postBytes)
            mstore(tempBytes, add(length, mload(tempBytes)))

            // Move the memory counter back from a multiple of 0x20 to the
            // actual end of the _preBytes data.
            mc := end
            // Stop copying when the memory counter reaches the new combined
            // length of the arrays.
            end := add(mc, length)

            for {
                let cc := add(_postBytes, 0x20)
            } lt(mc, end) {
                mc := add(mc, 0x20)
                cc := add(cc, 0x20)
            } {
                mstore(mc, mload(cc))
            }

            // Update the free-memory pointer by padding our last write location
            // to 32 bytes: add 31 bytes to the end of tempBytes to move to the
            // next 32 byte block, then round down to the nearest multiple of
            // 32. If the sum of the length of the two arrays is zero then add
            // one before rounding down to leave a blank 32 bytes (the length block with 0).
            mstore(0x40, and(
              add(add(end, iszero(add(length, mload(_preBytes)))), 31),
              not(31) // Round down to the nearest 32 bytes.
            ))
        }

        return tempBytes;
    }

    function concatStorage(bytes storage _preBytes, bytes memory _postBytes) internal {
        assembly {
            // Read the first 32 bytes of _preBytes storage, which is the length
            // of the array. (We don't need to use the offset into the slot
            // because arrays use the entire slot.)
            let fslot := sload(_preBytes.slot)
            // Arrays of 31 bytes or less have an even value in their slot,
            // while longer arrays have an odd value. The actual length is
            // the slot divided by two for odd values, and the lowest order
            // byte divided by two for even values.
            // If the slot is even, bitwise and the slot with 255 and divide by
            // two to get the length. If the slot is odd, bitwise and the slot
            // with -1 and divide by two.
            let slength := div(and(fslot, sub(mul(0x100, iszero(and(fslot, 1))), 1)), 2)
            let mlength := mload(_postBytes)
            let newlength := add(slength, mlength)
            // slength can contain both the length and contents of the array
            // if length < 32 bytes so let's prepare for that
            // v. http://solidity.readthedocs.io/en/latest/miscellaneous.html#layout-of-state-variables-in-storage
            switch add(lt(slength, 32), lt(newlength, 32))
            case 2 {
                // Since the new array still fits in the slot, we just need to
                // update the contents of the slot.
                // uint256(bytes_storage) = uint256(bytes_storage) + uint256(bytes_memory) + new_length
                sstore(
                    _preBytes.slot,
                    // all the modifications to the slot are inside this
                    // next block
                    add(
                        // we can just add to the slot contents because the
                        // bytes we want to change are the LSBs
                        fslot,
                        add(
                            mul(
                                div(
                                    // load the bytes from memory
                                    mload(add(_postBytes, 0x20)),
                                    // zero all bytes to the right
                                    exp(0x100, sub(32, mlength))
                                ),
                                // and now shift left the number of bytes to
                                // leave space for the length in the slot
                                exp(0x100, sub(32, newlength))
                            ),
                            // increase length by the double of the memory
                            // bytes length
                            mul(mlength, 2)
                        )
                    )
                )
            }
            case 1 {
                // The stored value fits in the slot, but the combined value
                // will exceed it.
                // get the keccak hash to get the contents of the array
                mstore(0x0, _preBytes.slot)
                let sc := add(keccak256(0x0, 0x20), div(slength, 32))

                // save new length
                sstore(_preBytes.slot, add(mul(newlength, 2), 1))

                // The contents of the _postBytes array start 32 bytes into
                // the structure. Our first read should obtain the `submod`
                // bytes that can fit into the unused space in the last word
                // of the stored array. To get this, we read 32 bytes starting
                // from `submod`, so the data we read overlaps with the array
                // contents by `submod` bytes. Masking the lowest-order
                // `submod` bytes allows us to add that value directly to the
                // stored value.

                let submod := sub(32, slength)
                let mc := add(_postBytes, submod)
                let end := add(_postBytes, mlength)
                let mask := sub(exp(0x100, submod), 1)

                sstore(
                    sc,
                    add(
                        and(
                            fslot,
                            0xffffffffffffffffffffffffffffffffffffffffffffffffffffffffffffff00
                        ),
                        and(mload(mc), mask)
                    )
                )

                for {
                    mc := add(mc, 0x20)
                    sc := add(sc, 1)
                } lt(mc, end) {
                    sc := add(sc, 1)
                    mc := add(mc, 0x20)
                } {
                    sstore(sc, mload(mc))
                }

                mask := exp(0x100, sub(mc, end))

                sstore(sc, mul(div(mload(mc), mask), mask))
            }
            default {
                // get the keccak hash to get the contents of the array
                mstore(0x0, _preBytes.slot)
                // Start copying to the last used word of the stored array.
                let sc := add(keccak256(0x0, 0x20), div(slength, 32))

                // save new length
                sstore(_preBytes.slot, add(mul(newlength, 2), 1))

                // Copy over the first `submod` bytes of the new data as in
                // case 1 above.
                let slengthmod := mod(slength, 32)
                let mlengthmod := mod(mlength, 32)
                let submod := sub(32, slengthmod)
                let mc := add(_postBytes, submod)
                let end := add(_postBytes, mlength)
                let mask := sub(exp(0x100, submod), 1)

                sstore(sc, add(sload(sc), and(mload(mc), mask)))

                for {
                    sc := add(sc, 1)
                    mc := add(mc, 0x20)
                } lt(mc, end) {
                    sc := add(sc, 1)
                    mc := add(mc, 0x20)
                } {
                    sstore(sc, mload(mc))
                }

                mask := exp(0x100, sub(mc, end))

                sstore(sc, mul(div(mload(mc), mask), mask))
            }
        }
    }

    function slice(
        bytes memory _bytes,
        uint256 _start,
        uint256 _length
    )
        internal
        pure
        returns (bytes memory)
    {
        require(_length + 31 >= _length, "slice_overflow");
        require(_bytes.length >= _start + _length, "slice_outOfBounds");

        bytes memory tempBytes;

        assembly {
            switch iszero(_length)
            case 0 {
                // Get a location of some free memory and store it in tempBytes as
                // Solidity does for memory variables.
                tempBytes := mload(0x40)

                // The first word of the slice result is potentially a partial
                // word read from the original array. To read it, we calculate
                // the length of that partial word and start copying that many
                // bytes into the array. The first word we copy will start with
                // data we don't care about, but the last `lengthmod` bytes will
                // land at the beginning of the contents of the new array. When
                // we're done copying, we overwrite the full first word with
                // the actual length of the slice.
                let lengthmod := and(_length, 31)

                // The multiplication in the next line is necessary
                // because when slicing multiples of 32 bytes (lengthmod == 0)
                // the following copy loop was copying the origin's length
                // and then ending prematurely not copying everything it should.
                let mc := add(add(tempBytes, lengthmod), mul(0x20, iszero(lengthmod)))
                let end := add(mc, _length)

                for {
                    // The multiplication in the next line has the same exact purpose
                    // as the one above.
                    let cc := add(add(add(_bytes, lengthmod), mul(0x20, iszero(lengthmod))), _start)
                } lt(mc, end) {
                    mc := add(mc, 0x20)
                    cc := add(cc, 0x20)
                } {
                    mstore(mc, mload(cc))
                }

                mstore(tempBytes, _length)

                //update free-memory pointer
                //allocating the array padded to 32 bytes like the compiler does now
                mstore(0x40, and(add(mc, 31), not(31)))
            }
            //if we want a zero-length slice let's just return a zero-length array
            default {
                tempBytes := mload(0x40)
                //zero out the 32 bytes slice we are about to return
                //we need to do it because Solidity does not garbage collect
                mstore(tempBytes, 0)

                mstore(0x40, add(tempBytes, 0x20))
            }
        }

        return tempBytes;
    }

    function toAddress(bytes memory _bytes, uint256 _start) internal pure returns (address) {
        require(_bytes.length >= _start + 20, "toAddress_outOfBounds");
        address tempAddress;

        assembly {
            tempAddress := div(mload(add(add(_bytes, 0x20), _start)), 0x1000000000000000000000000)
        }

        return tempAddress;
    }

    function toUint8(bytes memory _bytes, uint256 _start) internal pure returns (uint8) {
        require(_bytes.length >= _start + 1 , "toUint8_outOfBounds");
        uint8 tempUint;

        assembly {
            tempUint := mload(add(add(_bytes, 0x1), _start))
        }

        return tempUint;
    }

    function toUint16(bytes memory _bytes, uint256 _start) internal pure returns (uint16) {
        require(_bytes.length >= _start + 2, "toUint16_outOfBounds");
        uint16 tempUint;

        assembly {
            tempUint := mload(add(add(_bytes, 0x2), _start))
        }

        return tempUint;
    }

    function toUint32(bytes memory _bytes, uint256 _start) internal pure returns (uint32) {
        require(_bytes.length >= _start + 4, "toUint32_outOfBounds");
        uint32 tempUint;

        assembly {
            tempUint := mload(add(add(_bytes, 0x4), _start))
        }

        return tempUint;
    }

    function toUint64(bytes memory _bytes, uint256 _start) internal pure returns (uint64) {
        require(_bytes.length >= _start + 8, "toUint64_outOfBounds");
        uint64 tempUint;

        assembly {
            tempUint := mload(add(add(_bytes, 0x8), _start))
        }

        return tempUint;
    }

    function toUint96(bytes memory _bytes, uint256 _start) internal pure returns (uint96) {
        require(_bytes.length >= _start + 12, "toUint96_outOfBounds");
        uint96 tempUint;

        assembly {
            tempUint := mload(add(add(_bytes, 0xc), _start))
        }

        return tempUint;
    }

    function toUint128(bytes memory _bytes, uint256 _start) internal pure returns (uint128) {
        require(_bytes.length >= _start + 16, "toUint128_outOfBounds");
        uint128 tempUint;

        assembly {
            tempUint := mload(add(add(_bytes, 0x10), _start))
        }

        return tempUint;
    }

    function toUint256(bytes memory _bytes, uint256 _start) internal pure returns (uint256) {
        require(_bytes.length >= _start + 32, "toUint256_outOfBounds");
        uint256 tempUint;

        assembly {
            tempUint := mload(add(add(_bytes, 0x20), _start))
        }

        return tempUint;
    }

    function toBytes32(bytes memory _bytes, uint256 _start) internal pure returns (bytes32) {
        require(_bytes.length >= _start + 32, "toBytes32_outOfBounds");
        bytes32 tempBytes32;

        assembly {
            tempBytes32 := mload(add(add(_bytes, 0x20), _start))
        }

        return tempBytes32;
    }

    function equal(bytes memory _preBytes, bytes memory _postBytes) internal pure returns (bool) {
        bool success = true;

        assembly {
            let length := mload(_preBytes)

            // if lengths don't match the arrays are not equal
            switch eq(length, mload(_postBytes))
            case 1 {
                // cb is a circuit breaker in the for loop since there's
                //  no said feature for inline assembly loops
                // cb = 1 - don't breaker
                // cb = 0 - break
                let cb := 1

                let mc := add(_preBytes, 0x20)
                let end := add(mc, length)

                for {
                    let cc := add(_postBytes, 0x20)
                // the next line is the loop condition:
                // while(uint256(mc < end) + cb == 2)
                } eq(add(lt(mc, end), cb), 2) {
                    mc := add(mc, 0x20)
                    cc := add(cc, 0x20)
                } {
                    // if any of these checks fails then arrays are not equal
                    if iszero(eq(mload(mc), mload(cc))) {
                        // unsuccess:
                        success := 0
                        cb := 0
                    }
                }
            }
            default {
                // unsuccess:
                success := 0
            }
        }

        return success;
    }

    function equalStorage(
        bytes storage _preBytes,
        bytes memory _postBytes
    )
        internal
        view
        returns (bool)
    {
        bool success = true;

        assembly {
            // we know _preBytes_offset is 0
            let fslot := sload(_preBytes.slot)
            // Decode the length of the stored array like in concatStorage().
            let slength := div(and(fslot, sub(mul(0x100, iszero(and(fslot, 1))), 1)), 2)
            let mlength := mload(_postBytes)

            // if lengths don't match the arrays are not equal
            switch eq(slength, mlength)
            case 1 {
                // slength can contain both the length and contents of the array
                // if length < 32 bytes so let's prepare for that
                // v. http://solidity.readthedocs.io/en/latest/miscellaneous.html#layout-of-state-variables-in-storage
                if iszero(iszero(slength)) {
                    switch lt(slength, 32)
                    case 1 {
                        // blank the last byte which is the length
                        fslot := mul(div(fslot, 0x100), 0x100)

                        if iszero(eq(fslot, mload(add(_postBytes, 0x20)))) {
                            // unsuccess:
                            success := 0
                        }
                    }
                    default {
                        // cb is a circuit breaker in the for loop since there's
                        //  no said feature for inline assembly loops
                        // cb = 1 - don't breaker
                        // cb = 0 - break
                        let cb := 1

                        // get the keccak hash to get the contents of the array
                        mstore(0x0, _preBytes.slot)
                        let sc := keccak256(0x0, 0x20)

                        let mc := add(_postBytes, 0x20)
                        let end := add(mc, mlength)

                        // the next line is the loop condition:
                        // while(uint256(mc < end) + cb == 2)
                        for {} eq(add(lt(mc, end), cb), 2) {
                            sc := add(sc, 1)
                            mc := add(mc, 0x20)
                        } {
                            if iszero(eq(sload(sc), mload(mc))) {
                                // unsuccess:
                                success := 0
                                cb := 0
                            }
                        }
                    }
                }
            }
            default {
                // unsuccess:
                success := 0
            }
        }

        return success;
    }
}